

USING MATHEMATICA TO HELP ANALYZE FOURIER SERIES

As we continue our studies of Fourier series, included here a number of potentially useful tools to allow *Mathematica* to help in our work.

We can use Mathematica to help us evaluate Fourier coefficients. For instance, if we wish to find the Fourier series for $f(x) = x^2$ on $[-\pi, \pi]$, we would compute:

$$a_0 = \frac{1}{2\pi} \int_{-\pi}^{\pi} x^2 dx$$

via :

```
In[27]:= Integrate[x^2, {x, -π, π}] / (2 π)
```

```
Out[27]=  $\frac{\pi^2}{3}$ 
```

We can use symmetry to set all the b coefficients to zero, and the a coefficients are :

$$a_n = \frac{1}{\pi} \int_{-\pi}^{\pi} x^2 \cos(n x) dx$$

```
In[28]:= Integrate[x^2 Cos[n x], {x, -π, π}]
```

```
Out[28]=  $\frac{4 n \pi \text{Cos}[n \pi] + 2 (-2 + n^2 \pi^2) \text{Sin}[n \pi]}{n^3}$ 
```

Ok, that's helpful, and since we know that we are only interested in integer values of n, we can ignore the final term involving $\sin(n\pi)$ and figure out the expression for the nth coefficient. But in this and more complicated expressions, we can make use of two important Mathematica functions, Simplify and Assumptions, as :

```
In[29]:= Simplify[Integrate[x^2 Cos[n x], {x, -π, π}], Assumptions → n ∈ Integers]
```

```
Out[29]=  $\frac{4 (-1)^n \pi}{n^2}$ 
```

Isn't that much nicer? The symbol between "n" and "Integers" means "element of", and is made by typing the `ESC`el`ESC` keys.

We can combine many of the ideas we have learned already in the course and write a short program to help us compute Fourier series. The program below computes the Fourier series for the step function we used for our very first Fourier demonstration in class, but you can redefine $f[x]$ to

produce just about any series for functions satisfying Dirichlet's conditions.

```


Clear[f, a0, a, b, fourierseries]
(* The Which command allows us to define piecewise continuous functions *)
f[x_] := Which[-π < x < 0, 0, 0 < x < π, 1];
a0 = Integrate[f[x], {x, -π, π}]/(2 π);
a[n_] := a[n] = Simplify[Integrate[f[x] Cos[n x], {x, -π, π}], Assumptions -> n ∈ Integers]/π

b[n_] := b[n] = Simplify[Integrate[f[x] Sin[n x], {x, -π, π}], Assumptions -> n ∈ Integers]/π

fourierseries = a0 + Sum[a[m] Cos[m x] + b[m] Sin[m x], {m, 1, 11}];

Plot[fourierseries, {x, -π, π}]

```



If we want to see what the series looks like expanded, we can use the following:

```

In[47]:= Expand[fourierseries]
Out[47]=  $\frac{1}{2} + \frac{2 \sin[x]}{\pi} + \frac{2 \sin[3 x]}{3 \pi} + \frac{2 \sin[5 x]}{5 \pi} + \frac{2 \sin[7 x]}{7 \pi} + \frac{2 \sin[9 x]}{9 \pi} + \frac{2 \sin[11 x]}{11 \pi}$ 

```

And observe the explicit terms.

Suppose we have to integrate:

```

Simplify[Integrate[x^3 Cos[n x], {x, 0, π}], Assumptions -> n ∈ Integers]
Out[49]=  $\frac{6 + 3 (-1)^n (-2 + n^2 \pi^2)}{n^4}$ 

```

And we get an expression which while reduced, still seems as if it can be further simplified. Many times we will have expressions for Fourier coefficients which will take on different values depending on whether n is even or odd, (or in more complicated situations, whether n is evenly divisible by

4). For the integral above, we can make use of some other important *Mathematica* functions:

In[55]:= **Simplify[Integrate[x^3 Cos[n x], {x, 0, π}], Assumptions → n ∈ Integers]**

Out[55]=
$$\frac{3 (2 - 2 (-1)^n + (-1)^n n^2 \pi^2)}{n^4}$$

In[56]:= **Simplify[%, Mod[n, 2] == 0]**

Out[56]=
$$\frac{3 \pi^2}{n^2}$$

In[52]:= **Simplify[%%, Mod[n, 2] == 1]**

Out[52]=
$$\frac{12 - 3 n^2 \pi^2}{n^4}$$

The first line of output (line 51) provides the values of the coefficient when n is even; the second line gives the values for odd n . These commands introduce us to the use of "%", which means take the immediately previous output (hence, "%%" means take the next most recent output). The Mod command is familiar to those of you who have studied number theory. Mod[n, 2] determines the remainder of the quotient $n/2$. So in the first case, when we set the remainder of $n/2 = 0$, we determine the coefficients for even n . When we set Mod[n, 2] == 1, we are finding the values for odd n . Note that the Mod statement involves the double equal sign.